

## F04AGF – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

F04AGF calculates the approximate solution of a set of real symmetric positive-definite linear equations with multiple right-hand sides,  $AX = B$ , where  $A$  has been factorized by F03AEF.

### 2 Specification

```
SUBROUTINE F04AGF(N, IR, A, IA, P, B, IB, X, IX)
  INTEGER          N, IR, IA, IB, IX
  real           A(IA,N), P(N), B(IB,IR), X(IX,IR)
```

### 3 Description

To solve a set of real linear equations  $AX = B$  where  $A$  is symmetric positive-definite, the routine must be preceded by a call to F03AEF which computes a Cholesky factorization of  $A$  as  $A = LL^T$ , where  $L$  is lower triangular. The columns  $x$  of the solution  $X$  are found by forward and backward substitution in  $Ly = b$  and  $L^T x = y$ , where  $b$  is a column of the right-hand sides.

### 4 References

- [1] Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer-Verlag

### 5 Parameters

- 1: N — INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .
- 2: IR — INTEGER *Input*  
*On entry:*  $r$ , the number of right-hand sides.
- 3: A(IA,N) — *real* array *Input*  
*On entry:* the upper triangle of the  $n$  by  $n$  positive-definite symmetric matrix  $A$ , and the sub-diagonal elements of its Cholesky factor  $L$ , as returned by F03AEF.
- 4: IA — INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F04AGF is called.  
*Constraint:*  $IA \geq N$ .
- 5: P(N) — *real* array *Input*  
*On entry:* the reciprocals of the diagonal elements of  $L$ , as returned by F03AEF.
- 6: B(IB,IR) — *real* array *Input*  
*On entry:* the  $n$  by  $r$  right-hand side matrix  $B$ . See also Section 8.

- 7:** IB — INTEGER *Input*  
*On entry:* the first dimension of the array B as declared in the (sub)program from which F04AGF is called.  
*Constraint:* IB ≥ N .
- 8:** X(IX,IR) — *real* array *Output*  
*On exit:* the *n* by *r* solution matrix *X*. See also Section 8.
- 9:** IX — INTEGER *Input*  
*On entry:* the first dimension of the array X as declared in the (sub)program from which F04AGF is called.  
*Constraint:* IX ≥ N.

## 6 Error Indicators and Warnings

None.

## 7 Accuracy

The accuracy of the computed solutions depends on the conditioning of the original matrix. For a detailed error analysis see Wilkinson and Reinsch [1] page 39.

## 8 Further Comments

The time taken is approximately proportional to  $n^2r$ .

Unless otherwise stated in the Users' Note for your implementation, the routine may be called with the same actual array supplied for parameters B and X, in which case the solution vectors will overwrite the right-hand sides. However this is not standard Fortran 77, and may not work on all systems.

## 9 Example

To solve the set of linear equations  $AX = B$  where

$$A = \begin{pmatrix} 5 & 7 & 6 & 5 \\ 7 & 10 & 8 & 7 \\ 6 & 8 & 10 & 9 \\ 5 & 7 & 9 & 10 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 23 \\ 32 \\ 33 \\ 31 \end{pmatrix}.$$

### 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F04AGF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NMAX, IR, IA, IB, IX
      PARAMETER       (NMAX=8, IR=1, IA=NMAX, IB=NMAX, IX=NMAX)
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5, NOUT=6)
*      .. Local Scalars ..
      real            D1
      INTEGER          I, ID, IFAIL, J, N
*      .. Local Arrays ..
```

```

      real          A(IA,NMAX), B(IB,IR), P(NMAX), X(IX,IR)
*   .. External Subroutines ..
      EXTERNAL      F03AEF, F04AGF
*   .. Executable Statements ..
      WRITE (NOUT,*) 'F04AGF Example Program Results'
*   Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.GT.0 .AND. N.LE.NMAX) THEN
          READ (NIN,*) ((A(I,J),J=1,N),I=1,N), ((B(I,J),J=1,IR),I=1,N)
          IFAIL = 1
*
*   Cholesky decomposition
          CALL F03AEF(N,A,IA,P,D1,ID,IFAIL)
*
          IF (IFAIL.NE.0) THEN
              WRITE (NOUT,99999) 'Error in F03AEF. IFAIL =', IFAIL
          ELSE
*
*   Approximate solution of linear equations
              CALL F04AGF(N,IR,A,IA,P,B,IB,X,IX)
*
              WRITE (NOUT,*) ' Solution'
              DO 20 I = 1, N
                  WRITE (NOUT,99998) (X(I,J),J=1,IR)
20          CONTINUE
              END IF
          ELSE
              WRITE (NOUT,99999) 'N is out of range: N = ', N
          END IF
          STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (1X,8F9.4)
      END

```

## 9.2 Program Data

F04AGF Example Program Data

```

4
5   7   6   5
7  10   8   7
6   8  10   9
5   7   9  10
23  32  33  31

```

## 9.3 Program Results

F04AGF Example Program Results

```

Solution
1.0000
1.0000
1.0000
1.0000

```